
�TRNSYS Type 80 Heat Exchanger





	subroutine type80(TIME,XIN,OUT,T,DTDT,PAR,INFO,ICNTRL,*)

***********************************************************************

*  Type 80:  Heat Exchanger                                            

*                                                                      

*  The Type 80 Heat Exchanger models the performance of cataloged      

*  sensible heat exchanger (shell and tube, radiators, and similar    

*  geometries, where one fluid flows through a tube bank while        

*  another fluid flows normal to the tube bank.  Characteristic       

*  parameters calculated by TRNFIT are required.  Variations in the   

*  heat transfer coefficients are accounted for by calculating        

*  fluid transport properties with NEWFLUIDS.                         

*                                                                     

*  Updated 4/17/97  RJR                                               

*

***********************************************************************

	implicit none

	real time, t, dtdt, par,prop(7)

	character*2 units

	double precision xin, out

	integer info(15), icntrl,ierr,lur,luw,iform,luk

	integer iphase,ophase

	dimension xin(4), out(9), par(16)

	integer ni, np, nd, no, iopt, config

	integer ofltype,oflcomp,ifltype,iflcomp

	double precision C1,C2,C3,C4,C5,C6,Toi, 

     $         Tii,Too,Tio,Tom,Tim,Tsm,muom, 

     $         muos,muim,muis,mo,mi,Cmin,Cmax,Cr,Tomn,Timn,

     $         oscale,iscale,hAo,hAi,UAtot,NTU,eff,Qcalc,

     $         kom,kim,kos,kis,cpim,cpom,cpis,cpos,rhoom,rhoim,

     $         Prom,Pros,Prim,n,cpoconst,cpiconst,dPo,dPi,

     $         trans,C7,C8

	common /lunits/lur,luw,iform,luk

	external newfluids 



      

c    info(7) = -1 for first call ever of the component.

	if (info(7) .ge. 0) go to 100

	   ni    =  4 

C  Inputs:  1.  Outer fluid inlet temperature

C               2.  Outer fluid mass flow rate

C               3.  Inner fluid inlet temperature

C               4.  Inner fluid mass flow rate



	   np    =  16

C  Parameters:  1.  Configuration

C                      2.  Outer fluid type/number

C                      3.  Outer fluid composition

C                      4.  Outer fluid specific heat

C                      5.  Inner fluid type/number

C                      6.  Inner fluid composition

C                      7.  Inner fluid specific heat

C                      8.  Ratio of inner fluid mass flow rate over dynamic

C                            viscosity at transition

C                      9.  Outer hA coefficient

C                    10.  Outer Re exponent (heat transfer)

C                    11.  Inner hA coefficient (turbulent)

C                    12.  Inner hA coefficient (laminar)

C                    13.  Outer pressure drop coefficient

C                    14.  Outer Re exponent (pressure drop)

C                    15.  Inner pressure drop coefficient (turbulent)

C                    16.  Inner pressure drop coefficient (laminar)                



	   nd    =  0

	   no    =  9 

C  Outputs:  1.  Outer fluid outlet temperature

C                 2.  Outer fluid mass flow rate

C                 3.  Inner fluid outlet temperature

C                 4.  Inner fluid mass flow rate

C                 5.  Heat transfer rate

C                 6.  Heat exchanger effectiveness

C                 7.  Overall heat transfer coefficient-area product

C                 8.  Outer fluid pressure drop

C                 9.  Inner fluid pressure drop



	   info(6) = no



c  info(9) =1 just means to always call this component even if the

c      inputs aren't changing. See TRNSYS manual for details on info

c      array.



	   info(9) =  1



c  call to typeck insures number of inputs, outputs and parameters

c    in deck are same as required in this type.



	   call typeck(iopt,info,ni,np,nd)



c  since parameters remain constant through the entire simulation,

c    they can be set just once, inside this section of code.

 



100	continue



      config      =  nint(par(1))

      ofltype     =  nint(par(2))

      oflcomp   =  nint(par(3))

      cpoconst  =  dble(par(4))

      ifltype      =  nint(par(5))

      iflcomp    =  nint(par(6))

      cpiconst   =  dble(par(7))

      trans        =  dble(par(8))

      C1           =  dble(par(9))

      C2           =  dble(par(10))

      C3           =  dble(par(11))

      C4           =  dble(par(12))

      C5           =  dble(par(13))

      C6           =  dble(par(14))

      C7           =  dble(par(15))

      C8           =  dble(par(16))







      Toi    = xin(1)

      mo    = xin(2)

      Tii     = xin(3)

      mi     = xin(4)



      if ((mo .eq. 0.d0) .or. (mi .eq. 0.d0)) then

         Too = Toi

         Tio = Tii

         Qcalc = 0.d0

         eff = 0.d0

         UAtot = 0.d0

         hAo=0.d0

         hAi=0.d0



      else

C  Calculate mean temperatures



         Too = Toi

         Tio = Tii



150   continue



         Tom = (Toi + Too)/2.0d0

         Tim = (Tii + Tio)/2.0d0

         Tsm = (Tom + Tim)/2.0d0





C  Properties - outer fluid 

         call newfluids (ofltype,oflcomp,Tom,Tsm,cpom,cpos,

     $                  kom,kos,muom,muos,rhoom,ophase,*160)

         write(luw,*)  '***** Error linking with NEWFLUIDS *****'

         call mystop(1001)

         return 1

160   continue





C  Properties - inner fluid 

         call newfluids (ifltype,iflcomp,Tim,Tsm,cpim,cpis,

     $               kim,kis,muim,muis,rhoim,iphase,*170)

170   continue



C  Calculated heat exchanger performance



         Prom = muom*cpom/kom*1000

         Pros = muos*cpos/kos*1000

         hAo = C1*kom*(mo/muom)**C2*Prom**0.36*(Prom/Pros)**0.25



         Prim = muim*cpim/kim*1000



         if (iphase .eq. 1) then

            if ((mi/muim) .gt. trans) then

               if (Tii .lt. Toi) then

	           n = -0.5 

	        else

	           n = 0.0 

	        endif           

	        hAi = C3*kim*(mi/muim)**0.8*Prim**0.33

     $                        *((Tsm+273.2)/(Tim+273.2))**n

            else 

               if (Tii .lt. Toi) then

	           n = -0.25

               else

	           n = -0.08

	        endif

               hAi = C4*kim*(mi/muim)**0.33*Prim**0.33

     $                        *((Tsm+273.2)/(Tim+273.2))**n   

            endif

         else

            if ((mi/muim) .gt. trans) then

	        hAi = C3*kim*(mi/muim)**0.8*Prim**0.33

     $                              *(muis/muim)**(-0.14)

            else

	        hAi = C4*kim*(mi/muim)**0.33*Prim**0.33

     $                              *(muis/muim)**(-0.14)

            endif

         endif

 

         UAtot   =  1.0d0/(1.0d0/hAo+1.0d0/hAi)

	

C    Capacitance rates

         Cmin    =   min(mo*cpoconst,mi*cpiconst)

         Cmax    =   max(mo*cpoconst,mi*cpiconst)

         Cr      =   Cmin/Cmax



C    Number of transfer units 

         NTU     =  UAtot/Cmin



C    Parallel flow configuration	

         if (config .eq. 1) then

            eff =  (1.0d0-exp(-NTU*(1.0d0+Cr)))/(1+Cr)



C    Counterflow flow configuration	

         else if (config .eq. 2) then

            if (Cr .ge. 0.999) then

               eff = NTU/(1+NTU)

            else    

               eff =  (1.0d0-exp(-NTU*(1.0d0-Cr)))/

     $	              (1.0d0-Cr*exp(-NTU*(1.0d0-Cr)))

            endif



C    Crossflow configuration, single pass, both fluids unmixed

         else if (config .eq. 3) then

            eff = 1-exp((1.0d0/Cr)*NTU**0.22*(exp(-Cr*NTU**0.78)-1.0d0))



C    Crossflow configuration, single pass, one fluid mixed

C    Cmax mixed 

         else if (((config .eq. 4) .and. ((mo*cpom) .gt. (mi*cpim))).or.

     $          ((config .eq. 5) .and. ((mi*cpim) .gt. (mo*cpom)))) then

            eff = (1.0d0/Cr)*(1.0d0-exp(-Cr*(1.0d0-exp(-NTU)))) 

	

C    Cmin mixed

         else if (((config .eq. 4) .and. ((mo*cpom) .lt. (mi*cpim))).or.

     $          ((config .eq. 5) .and. ((mi*cpim) .lt. (mo*cpom)))) then

            eff = 1.0d0-exp((-1.0d0/Cr)*(1.0d0-exp(-Cr*NTU)))         



C    Crossflow configuration, single pass, both fluids mixed

         else if (config .eq. 6) then

            eff = ((1.0d0/(1.0d0-exp(-NTU)))+

     $             (Cr/(1.0d0-exp(-NTU*Cr)))-(1.0d0/NTU))**(-1)

     

C    Error

         else

            write(luw,*) '***** Error-invalid HX configuration *****'

            call mystop(1001)

            return 1 

	

         endif



C  Calculated heat transfer rate {kJ/hr}

         Qcalc  =  eff*Cmin*abs(Toi-Tii)



C  Calculated fluid outlet temperatures 

         if (Toi .gt. Tii) then

            Too = Toi-Qcalc/(mo*cpoconst)

            Tio = Tii+Qcalc/(mi*cpiconst)

         else

            Too = Toi+Qcalc/(mo*cpoconst)

            Tio = Tii-Qcalc/(mi*cpiconst)

         endif



C  New mean temperatures

         Tomn = (Toi + Too)/2.0d0

         Timn = (Tii + Tio)/2.0d0



C  Compare old and new mean temperatures, perform

C  calculations again if necessary



         oscale = max(abs(Toi - Too),1.0d-2)

         iscale = max(abs(Tii - Tio),1.0d-2)

         if ((abs(Tomn-Tom)/oscale .gt. 0.001) .or.

     $       (abs(Timn-Tim)/iscale .gt. 0.001)) go to 150



C  Pressure drops

         dPo = C5*(mo/muom)**C6*(mo**2/rhoom)

         if ((mi/muim) .gt. trans) then

            if (iphase .eq. 1) then

               if (Tii .gt. Toi) then

                  n=0.0

               else

                  n=0.12

               endif

               dPi = C7*(mi/muim)**(-0.25)*(mi**2/rhoim)

     $                            *((Tim+273.2)/(Tsm+273.2))**n

            else

               dPi = C7*(mi/muim)**(-0.25)*(mi**2/rhoim)

     $                             *(muis/muim)**(-0.14)

            endif

         else

            if (iphase .eq. 1) then

               if (Tii .gt. Toi) then

                  n=-0.30

               else

                  n=-0.45

               endif

               dPi = C8*(mi/muim)**(-1)*(mi**2/rhoim)

     $                            *((Tim+273)/(Tsm+273))**n

            else

               dPi = C8*(mi/muim)**(-1)*(mi**2/rhoim)

     $                              *(muis/muim)**(-0.14)

            endif

         endif



      endif



C  Outputs

	

      out(1) = Too  

      out(2) = mo 

      out(3) = Tio 

      out(4) = mi

      out(5) = Qcalc

      out(6) = eff 

      out(7) = UAtot 

      out(8) = dPo

      out(9) = dPi

	

      return 1

      end

�TRNSYS Type 85 Direct Expansion Cooling Coil





	subroutine type85(TIME,XIN,OUT,T,DTDT,PAR,INFO,ICNTRL,*)

***********************************************************************

*  Type 85:  Direct Expansion Cooling Coil                            

*                                                                     

*  The Type 85 Direct Expansion Cooling Coil models the performance 	

*  of cataloged DX coils.  The coil surface is treated as either      

*  totally dry or totally wet.  Characteristic parameters calculated   

*  by TRNFIT are required.  Variations in the heat transfer           

*  coefficients are accounted for by calculating refrigerant          

*  transport properties with REFTRANS and air transport properties    

*  with NEWFLUIDS										

*                                                                     

*  Updated 4/17/97  RJR                                               

*

***********************************************************************

	implicit none

	real time, t, dtdt, par,prop(7)

	real psydat(9),refprop(7)

	character*2 units

	parameter (units='SI')

	double precision xin, out

	integer info(15), icntrl, ierr, phase

	dimension xin(6), out(11), par(13)

	integer ni, np, nd, no, iopt,mode

	integer status,iflagr,nref,lur,luw,iform,luk

	double precision C1,C2,C3,C4,C5,C6,C7,wi,Tsm,

     $          eff,NTU,Cmin,UAtot,QS,QL,QT,Tdbo,Twbo,

     $          cpam,cpas,kam,kas,muam,muas,krl,murl,

     $          Pram,Pras,hAa,hAr,QSdry,QLdry,QTdry,

     $          nTdwo,Tdwbo,Tdpi,Twdbo,hTdpi,hTri,cpsat,

     $          hai,hrsi,rhostd,Vstd,Cf,UAhtot,capAirWet,

     $          mmin,hrso,Ri,hao,Rrat,Ro,hssi,hsso,

     $          Tsi,QTwet,hnew,airCap,cpv,Tcond,hscond,

     $          Twwbo,wo,Qmax,QLwet,hx,QSwet,nTwdbo,nTddbo,

     $          Tdbm,Tddbo,ma,mr,Twbi,Tdbi,rhoam,Tcndsr,

     $          cpa,Aface,Tdwo,dPa,dPad,dPaw,hri,hro,hfg,

     $          mrdry,mrwet,nmrdry,nmrwet,Tevap,xi,

     $          scdeg,shdeg,Pevap,wmax,hmax,fwet

	parameter (cpv=1.87,rhostd=1.184)

	common /lunits/lur,luw,iform,luk

	external fluids,newfluids,psych,reftrans 





c    info(7) = -1 for first call ever of the component.

         if (info(7) .ge. 0) go to 100

	

           ni    =  6

c  Inputs:      1:  inlet air dry bulb temperature

c                   2:  inlet air wet bulb temperature

c                   3:  air mass flow rate

c                   4:  evaporating temperature

c                   5:  condensing temperature

c                   6:  refrigerant mass flow rate

	

           np    =  13

c  Parameters:  1:  mode

c                      2:  air specific heat

c                      3:  refrigerant number

c                      4:  coil face area

c                      5:  degrees subcooling in the condenser

c                      6:  degrees superheating in the evaporator

c                      7:  air side hA coefficient

c                      8:  air side Re exponent

c                      9:  refrigerant side hA coefficient

c                    10:  air side pressure drop coefficient (dry)

c                    11:  air side pressure drop exponent (dry)

c                    12:  air side pressure drop coefficient (wet)

c                    13:  air side pressure drop exponent (wet)



           nd    =  0

           no    =  11 

c  Outputs:     1:  outlet air dry bulb temperature

c                    2:  outlet air wet bulb temperature

c                    3:  air mass flow rate

c                    4:  evaporation temperature

c                    5:  degrees superheat

c                    6:  refrigerant mass flow rate

c                    7:  sensible cooling rate

c                    8:  latent cooling rate

c                    9:  total cooling rate

c                  10:  air pressure drop

C                 11:  wet fraction of the coil (0 or 1)

	

         info(6) = no



c  info(9) =1 just means to always call this component even if the

c      inputs aren't changing. See TRNSYS manual for details on info

c      array.



         info(9) =  1



c  call to typeck ensures number of inputs, outputs and parameters

c    in deck are same as required in this type.



         call typeck(iopt,info,ni,np,nd)



c  since parameters remain constant through the entire simulation,

c    they can be set just once, inside this section of code.

 

100 continue



      mode     =  nint(par(1))

      cpa        =  dble(par(2))

      nref       =  nint(par(3))

      Aface    =  dble(par(4))

      scdeg     =  dble(par(5))

      shdeg     =  dble(par(6))

      C1         =  dble(par(7))

      C2         =  dble(par(8))

      C3         =  dble(par(9))

      C4         =  dble(par(10))

      C5         =  dble(par(11))

      C6         =  dble(par(12))

      C7         =  dble(par(13))

  



      Tdbi     = xin(1)

      Twbi    = xin(2)

      ma        = xin(3)

      Tevap   = xin(4)

      Tcndsr  = xin(5)

      mr        = xin(6)





      if ((ma .eq. 0.d0) .or. (mr .eq. 0.d0)) then

         Tdbo = Tdbi

         Twbo = Twbi

         QS   = 0.d0

         QL   = 0.d0

         QT   = 0.d0

	   

      else



         mrdry=mr

         mrwet=mr



C  Calculate humidity ratio of inlet air

         psydat(1) = 1.0

         psydat(2) = Tdbi

         psydat(3) = Twbi

         call psych(time,info,1,1,1,psydat,1,status,*105)

         write(luw,*) '***** Error linking with PSYCH *****'

         call mystop(1001)

         return 1

 105  continue

         wi = psydat(6)







C  Initial guesses for dry bulb temperature

         Tddbo = Tdbi - 20.d0



150   continue

         Tdbm = (Tdbi+Tddbo)/2.d0

         Tsm  = (Tdbm+Tevap)/2.d0



C  Refrigerant properties

C  Saturated liquid properties at evaporator temperature

         refprop(1)=Tevap

         refprop(5)=0.d0

         call fluids (units,refprop,nref,15,iflagr,*155)

         write(luw,*) '***** Error linking with FLUIDS *****'

         call mystop(1001)

         return 1

 155  continue

         Pevap=refprop(2)



         call reftrans(nref,Tevap,murl,krl,*157)

         write(luw,*) '***** Error linking with REFTRANS *****'

         call mystop(1001)

         return 1

 157  continue





C  Properties at evaporator outlet

         if (shdeg .eq. 0.d0) then

            refprop(1)=Tevap

            refprop(5)=1.0

            call fluids (units,refprop,nref,15,iflagr,*160)

 160	     continue

       

         else

            refprop(2)=Pevap

            refprop(1)=Tevap+shdeg

            call fluids (units,refprop,nref,21,iflagr,*165)

 165	     continue      

         endif



         hro=refprop(3)



C  Properties at condenser outlet

         refprop(1)=Tcndsr-scdeg

         refprop(5)=0.d0

         call fluids (units,refprop,nref,15,iflagr,*170)

 170  continue

         hri=refprop(3)



C  DRY COIL OPERATION

         if (mode .ne. 2) then 

C  Air properties

            call newfluids (2,0,Tdbm,Tsm,cpam,cpas,kam,kas,muam,

     $                 muas,rhoam,phase,*172)

            write(luw,*) '***** Error linking with NEWFLUIDS *****'

            call mystop(1001)

            return 1

 172     continue



C  Prandtl numbers

            Pram=muam*cpam/kam*1000

            Pras=muas*cpas/kas*1000   



C  hA products

C  Air

            hAa=C1*(ma/muam)**C2*Pram**0.36*

     $             (Pram/Pras)**0.25*kam



C  Refrigerant

            hAr=C3*krl*(mrdry/murl*(hro-hri))**0.45



            UAtot=1.d0/(1.d0/hAa+1.d0/hAr)



C  Capacitance rates

            Cmin=ma*cpa



C  Number of transfer units

            NTU=UAtot/Cmin



C  Effectiveness for capacitance rate ratio Cr=0

            eff=1.d0-exp(-NTU)



C  Heat Transfer

            QTdry=eff*Cmin*(Tdbi-Tevap)

            QSdry=QTdry

            QLdry=0.d0



C  Outlet States

            nTddbo=Tdbi-QTdry/(ma*cpa)



C  Mass flow rate calculation

            nmrdry=QTdry/(hro-hri)

            nmrdry=max(1.d-5,nmrdry)



C  Convergence

            if ((abs(Tddbo-nTddbo)/max(1.d0,Tddbo) .ge. 0.01) .or.

     $          (abs(mrdry-nmrdry)/max(1.d0,mrdry) .ge. 0.01)) then

               Tddbo = nTddbo

               mrdry = max(1.d-5,nmrdry)

               goto 150

            endif



            psydat(2)=Tddbo

            psydat(4)=1.d0

            call psych(time,info,1,2,1,psydat,2,status,*175)

175      continue

            wmax=psydat(6)



            psydat(2)=Tddbo

            psydat(6)=min(wmax,wi)

            call psych(time,info,1,4,1,psydat,2,status,*180)

 180     continue

            Tdwbo = psydat(3)



            dPad = C4*(ma**2/rhoam)*(ma/muam)**C5



            if (mode .eq. 1) then

               QT=QTdry

               QS=QSdry

               QL=QLdry

               Tdbo=Tddbo

               Twbo=Tdwbo

               dPa=dPad

               mr=mrdry

               fwet=0.d0

               goto 300

            endif

         endif





C  WET COIL OPERATION

         if (mode .ne. 1) then

C  Calculate entering dewpoint, make guess about leaving

C  dry bulb temperature and leaving liquid temperature 

            if (Tdbi .eq. Twbi) then

               Tdpi=Tdbi

            else

               psydat(2) = Tdbi

               psydat(3) = Twbi

               call psych(time,info,1,1,1,psydat,2,status,*185)

 185	 continue

               Tdpi = psydat(5)

            endif

            Twdbo = Tdbi - 20.d0



C  Mean temperatures

 200      continue

            Tdbm = (Tdbi+Twdbo)/2.d0

            Tsm = (Tdbm+Tevap)/2.d0



C  Saturation enthalpy at entering dewpoint

            psydat(2)=Tdpi

            psydat(4)=1.d0

            call psych(time,info,1,2,1,psydat,2,status,*205)

 205     continue

            hTdpi=psydat(7)



C  Saturation enthalpy at entering liquid temperature

            psydat(2)=Tevap

            call psych(time,info,1,2,1,psydat,2,status,*210)

 210     continue

            hTri=psydat(7)





C  Effective specific heat of saturated air

            cpsat=max((hTdpi-hTri),1.d-5)/max((Tdpi-Tevap),1.d-5)



C  Air enthalpy at inlet conditions

            psydat(2) = Tdbi

            psydat(3) = Twbi

            call psych(time,info,1,1,1,psydat,2,status,*215)

 215     continue

            hai=psydat(7)



C  Saturation enthalpy of air at inlet liquid temperature

            psydat(2) = Tevap

            psydat(4) = 1.d0

            call psych(time,info,1,2,1,psydat,2,status,*220)

 220     continue

            hrsi=psydat(7)



C  Air properties

            call newfluids (2,0,Tdbm,Tsm,cpam,cpas,kam,kas,muam,

     $                  muas,rhoam,phase,*221)

 221     continue



C  Prandtl numbers

            Pram=muam*cpam/kam*1000

            Pras=muas*cpas/kas*1000



C  Wet surface convection coefficient correction factor

            Vstd=ma/(rhostd*Aface)/60.0/0.3048

            Cf=0.626*Vstd**0.101





C  Conductance-heat transfer area products

            hAa=Cf*C1*(ma/muam)**C2*Pram**0.36*

     $            (Pram/Pras)**0.25*kam

  

            hAr=C3*krl*(mrwet/murl*(hro-hri))**0.45



C  Overall enthalpy transfer coefficient       

            UAhtot=1.d0/(cpsat/hAr+cpam/hAa)



C  Air-side capacity rate

            capAirWet=ma



C  Enthalpy-based capacitance rates

            mmin=capAirWet



C  Number of transfer units

            NTU=UAhtot/mmin



C  Effectiveness for capacitance rate ratio Cr=0

            eff=1-exp(-NTU)



C  Maximum heat transfer rate

            Qmax=mmin*abs(hai-hrsi)



C  Air enthalpy at outlet

            hao=hai-eff*Qmax/capAirWet        



C  Saturation enthalpy of air at exiting refrigerant temperature

            if (shdeg .gt. 0.d0) then

               refprop(2)=Pevap

               refprop(1)=Tevap+shdeg

               call fluids (units,refprop,nref,21,iflagr,*222)

 222        continue

               hrso=refprop(3)

            else

               hrso=hrsi

            endif



C  Heat transfer resistances

            Ri=1.d0/hAr

            Ro=1.d0/hAa

            Rrat=Ri/(Ri+cpam/cpsat*Ro)



C  Saturation enthalpy of air at entering and exit 

C  surface temperature

            hssi=hrso+Rrat*(hai-hrso)

            hsso=hrsi+Rrat*(hao-hrsi)



C  Total heat transfer rate for a totally wet coil

            QTwet=ma*(hai-hao)



C  Initial guess for surface temperature at air inlet

            Tsi=Tevap



C  Determine surface temperature corresponding to previously

C  calculated saturation enthalpy of air at entering

C  surface temperature

 225     continue

            psydat(2) = Tsi 

            psydat(4) = 1.d0

            call psych(time,info,1,2,1,psydat,2,status,*230)

 230     continue

            hnew=psydat(7)

            if (hnew .lt. hssi) then

               Tsi=Tsi + 0.1

               goto 225

            endif

   

C  Determine temperature effectiveness of air stream

            airCap=ma*(cpa+wi*cpv)

            NTU=hAa/airCap

            eff=1.d0-exp(-NTU)



C  Enthalpy of condensate

            hscond=hai-(hai-hao)/eff



C  First guess of condensate temperature

            Tcond = Tevap



C  Determine condensate temperature corresponding to

C  condensate enthalpy

 235     continue

            psydat(2) = Tcond

            psydat(4) = 1.d0

            call psych(time,info,1,2,1,psydat,2,status,*240)

 240     continue

            hnew=psydat(7)

            if (hnew .lt. hscond) then

               Tcond=Tcond+0.1

               goto 235

            endif



C  New calculated leaving dry bulb temperature for a wet coil

            nTwdbo=Tdbi-(Tdbi-Tcond)*eff



C  Calculate refrigerant mass flow rate

            nmrwet=QTwet/(hro-hri)

            nmrwet=max(1.d-5,nmrwet)



C  Check convergence of leaving dry bulb temperature and

C  leaving liquid temperature

            if ((abs(Twdbo-nTwdbo)/max(1.d0,Twdbo) .ge. 0.01) .or.

     $         (abs(mrwet-nmrwet)/max(1.d0,mrwet) .ge. 0.01)) then

               Twdbo=nTwdbo

               mrwet=max(1.d-5,nmrwet)

               goto 200

            endif



C  Calculate leaving wet bulb temperature and leaving   

C  humidity ratio

            psydat(2)=Twdbo

            psydat(4)=1.d0

            call psych(time,info,1,2,1,psydat,2,status,*243)

243      continue

            hmax=psydat(7)       





            psydat(2)=Twdbo

            psydat(7)=min(hmax,hao)

            call psych(time,info,1,5,1,psydat,2,status,*245)

 245     continue

            Twwbo=psydat(3)

     

            if (Twwbo .gt. Twdbo) then

               psydat(2)=Twdbo

               psydat(3)=Twdbo

               call psych(time,info,1,1,1,psydat,2,status,*250)          

 250	        continue

            endif



            wo=psydat(6)



            psydat(2)=Twdbo

            psydat(4)=1.d0

            call psych(time,info,1,2,1,psydat,2,status,*255)

255      continue

            wmax=psydat(6)



C  Calculate enthalpy at fictitious state x

            psydat(2)=Tdbi

            psydat(6)=min(wmax,wo) 

            call psych(time,info,1,4,1,psydat,2,status,*260)

 260     continue

            hx=psydat(7)



C  Calculate latent and sensible heat transfers

            QLwet=ma*(hai-hx)

            QSwet=ma*(hx-hao)



            dPaw = C6*(ma**2/rhoam)*(ma/muam)**C7



            if (mode .eq. 2) then

               QT=QTwet

               QS=QSwet

               QL=QLwet

               Tdbo=Twdbo

               Twbo=Twwbo

               dPa=dPaw

               mr=mrwet

	  fwet=1.d0

               goto 300

            endif

         endif



C  Decide whether coil should be considered as totally dry

C  totally wet





         if (Tdpi .le. Tevap) then

C  Coil is totally dry           

            QT=QTdry

            QS=QSdry

            QL=QLdry

            Tdbo=Tddbo

            Twbo=Tdwbo

            dPa=dPad

            mr=mrdry

            fwet=0.d0



         else

            if (Tdpi .le. Tsi) then

C  Dewpoint of inlet air is between inlet surface temperature and

C  refrigerant temperature; consider coil to be totally dry or

C  totally wet, whichever has the greatest total heat transfer

               if (QTdry .gt. QTwet) then

                  QT=QTdry

                  QS=QSdry

                  QL=QLdry

                  Tdbo=Tddbo

                  Twbo=Tdwbo            

                  dPa=dPad

                  mr=mrdry

                  fwet=0.d0

               else

                  QT=QTwet

                  QS=QSwet

                  QL=QLwet

                  Tdbo=Twdbo

                  Twbo=Twwbo

                  dPa=dPaw

                  mr=mrwet

	     fwet=1.d0

               endif



            else

C  Coil is totally wet

               QT=QTwet

               QS=QSwet

               QL=QLwet

               Tdbo=Twdbo

               Twbo=Twwbo

               dPa=dPaw

               mr=mrwet

	  fwet=1.d0

            endif

         endif

      endif



300 continue



C  Outputs

      out(1)   = Tdbo

      out(2)   = Twbo

      out(3)   = ma

      out(4)   = Tevap

      out(5)   = shdeg

      out(6)   = mr

      out(7)   = QS 

      out(8)   = QL

      out(9)   = QT

      out(10) = dPa

      out(11) = fwet



      return 1

      end

�TRNSYS Type 94 Chilled Water Cooling Coil





	subroutine type94(TIME,XIN,OUT,T,DTDT,PAR,INFO,ICNTRL,*)

***********************************************************************

*  Type 94:  Chilled Water Cooling Coil                               

*                                                                     

*  The Type 94 Chilled Water Cooling Coil models the performance of 	

*  cataloged cooling coils.  The coil surface can be treated as       

*  partially wet.  Characteristic parameters calculated by TRNFIT are 

*  required.  Variations in the heat transfer coefficients are        

*  accounted for by calculating fluid transport properties with       

*  NEWFLUIDS.						                                    

*                                                                     

*  Updated 4/17/97  RJR                                               

*												 

***********************************************************************

	implicit none

	real time, t, dtdt, par,prop(7)

	real psydat(9)

	character*2 units

	double precision xin, out

	integer info(15), icntrl, ierr

	integer lur,luw,iform,luk,it2,itmax

	parameter (itmax=100)

	dimension xin(5), out(11), par(18)

	integer ni, np, nd, no, iopt,turb

	integer liqtype,liqcomp,status,mode

	double precision Aface,C1,C2,C3,C4,C5,C6,C1dry,C1wet,

     $          C2dry,C2wet,C3dry,C3wet,C4dry,C4wet,C5dry,C5wet,

     $          cpa,cpw,dPad,dPaw,dPwd,dPww,

     $          fwet,ma,mw,QLwet,QSwet,Qdry,QTwet,small,

     $          Tdbi,Tdbo,Tddbo,Tdpi,Tdwbo,TLiqBnd,

     $          TNewLiqBnd,TSurfBnd,Twbi,Twbo,Twi,Two,

     $          f1,f2,err1,err2,err,df,tiny,Tso

	common /lunits/lur,luw,iform,luk

	parameter (small=1.d-5,tiny=1.d-9)

	external newfluids,psych,drycoil,wetcoil 



c    info(7) = -1 for first call ever of the component.

         if (info(7) .ge. 0) go to 10

	

            ni    =  5

c  Inputs:  1:  inlet air dry bulb temperature

c               2:  inlet air wet bulb temperature

c               3:  air mass flow rate

c               4:  inlet liquid temperature

c               5:  liquid mass flow rate

	

            np    =  18

c  Parameters:  1:  mode

c                      2:  air specific heat

c                      3:  liquid type

c                      4:  liquid composition (e.g. % EG)

c                      5:  liquid specific heat

c                      6:  coil face area

c                      7:  turbulators? 1-yes else-no

c                      8:  air side hA coefficient (dry)

c                      9:  air side hA coefficient (wet)

c                    10:  air side Re exponent (dry)

c                    11:  air side Re exponent (wet)

c                    12:  liquid side hA coefficient (dry)

c                    13:  liquid side hA coefficient (wet)

c                    14:  air side pressure drop coefficient (dry)

c                    15:  air side pressure drop coefficient (wet)

c                    16:  air side pressure drop exponent (dry)

c                    17:  air side pressure drop exponent (wet)

c                    18:  liquid side pressure drop coefficient

	

         nd    =  0

         no    = 11 

c  Outputs:     1:  outlet air dry bulb temperature

c                    2:  outlet air wet bulb temperature

c                    3:  air mass flow rate

c                    4:  outlet liquid temperature

c                    5:  liquid mass flow rate

c                    6:  sensible cooling rate

c                    7:  latent cooling rate

c                    8:  total cooling rate

c                    9:  air pressure drop

c                  10:  liquid pressure drop

c                  11:  fraction of the coil surface that is wet

	

         info(6) = no



c  info(9) =1 just means to always call this component even if the

c      inputs aren't changing. See TRNSYS manual for details on info

c      array.



         info(9) =  1



c  call to typeck insures number of inputs, outputs and parameters

c    in deck are same as required in this type.



         call typeck(iopt,info,ni,np,nd)



c  since parameters remain constant through the entire simulation,

c    they can be set just once, inside this section of code.



 

10    continue



      mode         =  nint(par(1)) 

      cpa            =  dble(par(2))

      liqtype       =  nint(par(3))

      liqcomp     =  nint(par(4))

      cpw           =  dble(par(5))

      Aface         =  dble(par(6))

      turb            =  nint(par(7))

      C1dry        =  dble(par(8))

      C1wet        =  dble(par(9))

      C2dry        =  dble(par(10))

      C2wet        =  dble(par(11))

      C3dry        =  dble(par(12))

      C3wet        =  dble(par(13))

      C4dry        =  dble(par(14))

      C4wet        =  dble(par(15))

      C5dry        =  dble(par(16))

      C5wet        =  dble(par(17))

      C6             =  dble(par(18))

        



      Tdbi    = xin(1)

      Twbi   = xin(2)

      ma       = xin(3)

      Twi     = xin(4)

      mw      = xin(5)



      if ((ma .eq. 0.d0) .or. (mw .eq. 0.d0)) then

         Tdbo = Tdbi

         Twbo = Twbi

         Two  = Twi

         Qdry   = 0.d0

         QTwet  = 0.d0

         QSwet  = 0.d0

         QLwet  = 0.d0

         dPaw   = 0.d0

         dPad   = 0.d0

         dPww   = 0.d0

         dPwd   = 0.d0

         fwet   = 0.d0



      else



         psydat(1) = 1.0

         psydat(2) = Tdbi

         psydat(3) = Twbi

         call psych(time,info,1,1,1,psydat,2,status,*20)

         write(luw,*) '***** Error linking with PSYCH *****'

         call mystop(1001)

         return 1

 20    continue

         Tdpi = psydat(5)



C  TOTALLY DRY COIL

         if (mode .ne. 2) then

            f1=0.d0

            C1=C1dry

            C2=C2dry

            C3=C3dry

            C4=C4dry

            C5=C5dry



            TLiqBnd=Twi



            call drycoil(time,info,ma,mw,Tdbi,

     $               Twbi,TLiqBnd,C1,C2,C3,C4,C5,C6,

     $               Qdry,Tddbo,Tdwbo,Two,dPad,dPwd,

     $               cpa,cpw,turb,liqtype,liqcomp,Tso,*21)

            write(luw,*) '***** Error linking with DRYCOIL *****'

            call mystop(1001)

            return 1

 21       continue



            Tdbo=Tddbo

            Twbo=Tdwbo

            QSwet=0.d0

            QLwet=0.d0

            QTwet=0.d0

            dPaw=0.d0

            dPww=0.d0 

            fwet=0.d0      



            if (mode .eq. 1) goto 70



            err1=Tso-Tdpi

            if (Tso .gt. Tdpi) then

               fwet=0.d0

               goto 70

            endif

         endif        



C  TOTALLY WET COIL

         if (mode .ne. 1) then

            f2=1.d0

            C1=C1wet

            C2=C2wet

            C3=C3wet

            C4=C4wet

            C5=C5wet



            call wetcoil(time,info,ma,mw,Tdbi,Twbi,

     $               Twi,C1,C2,C3,C4,C5,C6,QTwet,QSwet,

     $               QLwet,Tdbo,Twbo,TNewLiqBnd,dPaw,dPww,

     $               cpa,cpw,Aface,TSurfBnd,turb,liqtype,

     $               liqcomp,*24)

 24       continue



            Qdry=0.d0

            dPad=0.d0

            dPwd=0.d0

            fwet=1.d0



            if (mode .eq. 2) goto 70



            err2=TSurfBnd-Tdpi

            if (TSurfBnd .lt. Tdpi) then

               fwet=1.d0

               goto 70

            endif



         endif



         Two=Tdbi

         TLiqBnd=Tdpi

         fwet=(TLiqBnd-Twi)/(Two-Twi)

           

 30    continue



         TLiqBnd=Twi+fwet*(Two-Twi)

         it2=0



 50    continue



         it2=it2+1

         C1=max((1.d0-fwet)*C1dry,tiny*C1dry)

         C2=C2dry

         C3=max((1.d0-fwet)*C3dry,tiny*C3dry)

         C4=max((1.d0-fwet)*C4dry,tiny*C4dry)

         C5=C5dry



         call drycoil(time,info,ma,mw,Tdbi,

     $               Twbi,TLiqBnd,C1,C2,C3,C4,C5,C6,

     $               Qdry,Tddbo,Tdwbo,Two,dPad,dPwd,

     $               cpa,cpw,turb,liqtype,liqcomp,Tso,*52)

 52    continue



         C1=max(fwet*C1wet,tiny*C1wet)

         C2=C2wet

         C3=max(fwet*C3wet,tiny*C3wet)

         C4=max(fwet*C4wet,tiny*C4wet)

         C5=C5wet



         call wetcoil(time,info,ma,mw,Tddbo,Tdwbo,

     $               Twi,C1,C2,C3,C4,C5,C6,QTwet,QSwet,

     $               QLwet,Tdbo,Twbo,TNewLiqBnd,dPaw,dPww,

     $               cpa,cpw,Aface,TSurfBnd,turb,liqtype,

     $               liqcomp,*54)

 54    continue



         if (TNewLiqBnd .ne. TLiqBnd) then

            if (abs(TNewLiqBnd-TLiqBnd) .le. small) goto 60

            TLiqBnd=TNewLiqBnd 

            if (it2 .ge. itmax) goto 60

            goto 50



         endif



60     continue



         err=TSurfBnd-Tdpi



         if ((err*err1) .gt. 0.d0) then

            err1=err

            df=abs(f1-fwet)

            f1=fwet

         else

           err2=err

           df=abs(f2-fwet)

            f2=fwet

         endif



         if (df .gt. small) then

            fwet=-1.d0*(err1-f1*(err1-err2)/(f1-f2))

     $                /((err1-err2)/(f1-f2))

            fwet=max(0.d0,min(fwet,1.d0))

            goto 30

         endif



      endif

70 continue



      out(1)  = Tdbo

      out(2)  = Twbo

      out(3)  = ma

      out(4)  = Two

      out(5)  = mw

      out(6)  = Qdry+QSwet

      out(7)  = QLwet

      out(8)  = Qdry+QTwet

      out(9)  = dPad+dPaw

      out(10) = dPwd+dPww

      out(11) = fwet



      return 1

      end

�TRNSYS Type 95 Chilled Water Cooling Coil





	subroutine type95(TIME,XIN,OUT,T,DTDT,PAR,INFO,ICNTRL,*)

***********************************************************************

*  Type 95:  Chilled Water Cooling Coil                               

*                                                                     

*  The Type 95 Chilled Water Cooling Coil models the performance of 	

*  cataloged cooling coils.  The coil surface is treated as either    

*  totally dry or totally wet.  Characteristic parameters calculated   

*  by TRNFIT are required.  Variations in the heat transfer           

*  coefficients are accounted for by calculating fluid transport      

*  properties with NEWFLUIDS.						                    

*                                                                     

*  Updated 4/17/97  RJR                                               

*

***********************************************************************

	implicit none

	real time, t, dtdt, par,prop(7)

	real psydat(9)

	character*2 units

	double precision xin, out

	integer info(15), icntrl, ierr

	dimension xin(5), out(11), par(15)

	integer ni, np, nd, no, iopt, turb, mode

	integer liqtype,liqcomp,status,lur,luw,iform,luk

	double precision Aface,C1,C2,C3,C4,C5,C6,C7,C8,

     $          cpa,cpw,dPa,dPad,dPaw,dPw,dPwd,dPww,ma,mw,

     $          QL,QLwet,QS,QSwet,QT,Qdry,QTwet,Tdbi,Tdbo,

     $          Tddbo,Tdpi,Tdwbo,Tdwo,Twbi,Twbo,Twdbo,Twi,

     $          Two,Twwo,Tsi,Twwbo,fwet,Tso

	common /lunits/lur,luw,iform,luk

	external drycoil,psych,wetcoil 



c    info(7) = -1 for first call ever of the component.

         if (info(7) .ge. 0) go to 100

	

            ni    =  5

c  Inputs:      1:  inlet air dry bulb temperature

c                   2:  inlet air wet bulb temperature

c                   3:  air mass flow rate

c                   4:  inlet liquid temperature

c                   5:  liquid mass flow rate

	

            np    =  15 

c  Parameters:  1:  mode

c                      2:  air specific heat

c                      3:  liquid type

c                      4:  liquid composition (e.g. % EG)

c                      5:  liquid specific heat

c                      6:  coil face area

c                      7:  turbulators?  1-yes  else-no

c                      8:  air side UA coefficient

c                      9:  air side Re exponent

c                    10:  liquid side UA coefficient

c                    11:  air side pressure drop coefficient (dry)

c                    12:  air side pressure drop exponent (dry)

c                    13:  air side pressure drop coefficient (wet)

c                    14:  air side pressure drop exponent (wet)

c                    15:  liquid side pressure drop coefficient

	

            nd    =  0

            no    = 11 

c  Outputs:     1:  outlet air dry bulb temperature

c                    2:  outlet air wet bulb temperature

c                    3:  air mass flow rate

c                    4:  outlet liquid temperature

c                    5:  liquid mass flow rate

c                    6:  sensible cooling rate

c                    7:  latent cooling rate

c                    8:  total cooling rate

c                    9:  air pressure drop

c                  10:  liquid pressure drop

c                  11:  wet fraction of the coil (0 or 1)

	

            info(6) = no



c  info(9) =1 just means to always call this component even if the

c      inputs aren't changing. See TRNSYS manual for details on info

c      array.



            info(9) =  1



c  call to typeck insures number of inputs, outputs and parameters

c    in deck are same as required in this type.



            call typeck(iopt,info,ni,np,nd)



c  since parameters remain constant through the entire simulation,

c    they can be set just once, inside this section of code.

 

100   continue



      mode      =  nint(par(1))

      cpa         =  dble(par(2))

      liqtype    =  nint(par(3))

      liqcomp  =  nint(par(4))

      cpw        =  dble(par(5))

      Aface      =  dble(par(6))

      turb        =  nint(par(7))

      C1          =  dble(par(8))

      C2          =  dble(par(9))

      C3          =  dble(par(10))

      C4          =  dble(par(11))

      C5          =  dble(par(12))

      C6          =  dble(par(13))

      C7          =  dble(par(14))

      C8          =  dble(par(15))



        

      Tdbi    = xin(1)

      Twbi   = xin(2)

      ma       = xin(3)

      Twi     = xin(4)

      mw     = xin(5)





      if ((ma .eq. 0.d0) .or. (mw .eq. 0.d0)) then

         Tdbo = Tdbi

         Twbo = Twbi

         Two  = Twi

         QS   = 0.d0

         QL   = 0.d0

         QT   = 0.d0

         fwet = 0.d0

	   

      else



         psydat(1) = 1.0

         psydat(2) = Tdbi

         psydat(3) = Twbi

         call psych(time,info,1,1,1,psydat,1,status,*101)

         write(luw,*) 'Error - psych procedure not found'

         call mystop(1001)

         return 1

 101  continue

         Tdpi = psydat(5)



         call drycoil(time,info,ma,mw,Tdbi,

     $               Twbi,Twi,C1,C2,C3,C4,C5,C8,

     $               Qdry,Tddbo,Tdwbo,Tdwo,dPad,dPwd,

     $               cpa,cpw,turb,liqtype,liqcomp,Tso,*110)

         write(luw,*) '***** Error linking with DRYCOIL *****'

         call mystop(1001)

         return 1

 110  continue



         call wetcoil(time,info,ma,mw,Tdbi,Twbi,

     $               Twi,C1,C2,C3,C6,C7,C8,QTwet,QSwet,

     $               QLwet,Twdbo,Twwbo,Twwo,dPaw,dPww,

     $               cpa,cpw,Aface,Tsi,turb,liqtype,

     $               liqcomp,*115)

         write(luw,*) '***** Error linking with WETCOIL *****'

         call mystop(1001)

         return 1

 115  continue

        

C  Decide whether coil should be considered as totally dry

C  totally wet



         if ((mode .ne. 1) .and. (mode .ne. 2)) then

            if (Tdpi .le. Twi) then

C  Coil is totally dry           

               QT=Qdry

               QS=Qdry

               QL=0.d0

               Tdbo=Tddbo

               Twbo=Tdwbo

               Two=Tdwo

               dPa=dPad

               dPw=dPwd

               fwet=0.d0       



            else

               if (Tdpi .le. Tsi) then

C  Dewpoint of inlet air is between inlet surface temperature and

C  inlet water temperature; consider coil to be totally dry or

C  totally wet, whichever has the greater total heat transfer

                  if (Qdry .gt. QTwet) then

                     QT=Qdry

                     QS=Qdry

                     QL=0.d0

                     Tdbo=Tddbo

                     Twbo=Tdwbo

                     Two=Tdwo            

                     dPa=dPad

                     dPw=dPwd

                     fwet=0.d0

                  else

                     QT=QTwet

                     QS=QSwet

                     QL=QLwet

                     Tdbo=Twdbo

                     Twbo=Twwbo

                     Two=Twwo

                     dPa=dPaw

                     dPw=dPww

                     fwet=1.d0

                  endif



               else

C  Coil is totally wet

                  QT=QTwet

                  QS=QSwet

                  QL=QLwet

                  Tdbo=Twdbo

                  Twbo=Twwbo

                  Two=Twwo

                  dPa=dPaw

                  dPw=dPww

                  fwet=1.d0

               endif

            endif



         else if (mode .eq. 1) then

            QT=Qdry

            QS=Qdry

            QL=0.d0

            Tdbo=Tddbo

            Twbo=Tdwbo

            Two=Tdwo

            dPa=dPad

            dPw=dPwd

            fwet=0.d0



         else if (mode .eq. 2) then

            QT=QTwet

            QS=QSwet

            QL=QLwet

            Tdbo=Twdbo

            Twbo=Twwbo

            Two=Twwo

            dPa=dPaw

            dPw=dPww

            fwet=1.d0

         endif



      endif





C  Outputs

      out(1)   = Tdbo

      out(2)   = Twbo

      out(3)   = ma

      out(4)   = Two

      out(5)   = mw

      out(6)   = QS 

      out(7)   = QL

      out(8)   = QT

      out(9)   = dPa

      out(10) = dPw

      out(11) = fwet

      

      return 1

      end

�TRNSYS Type 99 Fitter and Subroutine ERRCALC





       subroutine type99(time,xin,out,t,dtdt,par,info,icntrl,*)

	use MSIMSLMD

	use d2arrays

***********************************************************************

*  Type 99:  Fitter                                                   

*                                                                     

*  The Type 99 Fitter allows TRNSYS and the IMSL routines to work 	

*  together within TRNFIT.  Type 99 reads the Input Data File, calls	

*  the IGV subroutine, calls the IMSL routine, and writes part of     

*  of the Output File.

*					

*  Updated 4/17/97  RJR                                               

* 

***********************************************************************

	implicit none

	real time,t,dtdt,par,prop(7)

	character*2 units

	character*4 typestring

	doubleprecision xin,out

	doubleprecision fscale,err,rparam(7),fvalue

	doubleprecision rhostd

	parameter (rhostd=1.184d0)

	integer lur,luw,iform,luk,ni,no,np,nout,LUin,LUout

	integer info(15),icntrl,ierr,outdata,outtype,maxiter

	integer ibtype,flag,nd,tnumber

	dimension xin(0),out(0),par(8)

	integer nfit,ndata,i,j,k,iparam(7)

	integer iersvr,ipact,isact,iopt,nptype,nitype,notype,maxchar

	parameter (iersvr=0,ipact=0,isact=0)

	common /lunits/lur,luw,iform,luk

	common /dtype/nptype,nitype,notype

	common /parrray/ndata,nout,outdata,outtype,LUout,

     $                LUin,maxchar

	common /etc/flag,tnumber

	external errcalc



      maxchar=15



      if (info(7) .ge. 0) goto 120



      ni      = 0



      np      = 8

C  Parameters:	1.  Number of characteristic parameters to be fit

C                     2.  TRNSYS Type output number with which parameters are to be fit

C                     3.  Number of parameters required by the TRNSYS Type

C                     4.  Number of inputs required by the TRNSYS Type

C                     5.  Number of outpus returned by the TRNSYS Type

C                     6.  Maximum number of iterations

C                     7.  Logical unit number of the Input Data File

C                     8.  Logical unit number of the Output File



      no      = 0



      info(6) = no



      info(9) = 1



      call typeck(iopt,info,ni,np,nd)



      nfit        = nint(par(1))

      outtype  = nint(par(2))

      nptype   = nint(par(3))

      nitype    = nint(par(4))

      notype   = nint(par(5))

      maxiter  = nint(par(6))

      LUin     = nint(par(7))  

      LUout   = nint(par(8))



      allocate (xguess(nfit))

      allocate (xlb(nfit))

      allocate (xub(nfit))

      allocate (xscale(nfit))

      allocate (xsol(nfit))

      allocate (partype(nptype))





C  ***** READ INPUT DATA FILE *****

      read(LUin,*) typestring,tnumber



      read(LUin,*) ndata

	

      do i=1,nptype

         read(LUin,*) partype(i)

      enddo



      allocate (fitpar(nfit))

      read(LUin,*) (fitpar(i),i=1,nfit)



      read(LUin,*) nout,outdata



      allocate (outhead(nout))

 	   

      allocate (input(ndata,nitype))

      allocate (output(ndata,nout))



      do i=1,ndata

         read(LUin,*) (input(i,j),j=1,nitype),(output(i,k),k=1,nout)

      enddo

	

      allocate (typeout(nout))

      read(LUin,*)  (typeout(i),i=1,nout) 



      allocate (calc(ndata,notype))



      call igv (*100)

      write(luw,*) '***** Error linking with IGV *****'

      call mystop(1001)

      return 1

100 continue



      do i=1,nfit

         partype(fitpar(i))=xguess(i)

      enddo



C  ***** PARAMETER FITTING *****

      ibtype=0



      do i=1,nfit

         xscale(i)=1.d0

      enddo



      fscale=1.d0



      call du4inf (iparam,rparam)

      iparam(3)=maxiter

      iparam(4)=2*maxiter

      iparam(5)=2*maxiter



      flag=0

      call dbconf (errcalc,nfit,xguess,ibtype,xlb,xub,xscale,fscale,

     $             iparam,rparam,xsol,fvalue)



C  ***** WRITE TO OUTPUT FILE *****



      flag=1

      call errcalc(nfit,xsol,err)



      write(LUout,*) ' ADDITIONAL RESULTS'

      write(LUout,*)

      write(LUout,110) '  The sum of squared normalized errors is ',

     $                                                        fvalue

      write(LUout,*) ' The number of iterations is ',iparam(3)

      write(LUout,*) ' The number of function evaluations is ',iparam(4)

      write(LUout,*) ' The number of gradient evaluations is ',iparam(5)

      write(LUout,*)

      write(LUout,*) '****************************************'



110 format (A,1X,E15.4)

120 CONTINUE

      RETURN 1

      end



C ===========================================================

	subroutine errcalc (nfit,cval,err)

	use d2arrays

***********************************************************************

*  Subroutine ERRCALC                                                 

*                                                                     

*  Subroutine ERRCALC calculates the objective error function to be   

*  to be minimized by the IMSL routine.  When the parameter           

*  estimation is completed, it calls subroutine RESULTS to write the  

*  Output File.											

*                                                                     

*  Updated 4/17/97  RJR                                               

*

***********************************************************************

	implicit none

	integer flag,i,ndata

	integer lur,luw,iform,luk,nptype,notype

	integer nitype,nfit,outtype,outdata,nout,LUout

	integer LUin,maxchar,tnumber

	doubleprecision err

	doubleprecision cval(nfit)

	common /lunits/lur,luw,iform,luk

	common /dtype/nptype,nitype,notype

	common /parrray/ndata,nout,outdata,outtype,LUout,

     $                LUin,maxchar

	common /etc/flag,tnumber



      err=0.d0

      call runtype(ndata,nfit,cval,*300)

      write(luw,*) '***** Error linking with IGV *****'

      call mystop(1001)

      return 1

300 continue



      do i=1,ndata

         err=err+((calc(i,outtype)-output(i,outdata))

     $                              /output(1,outdata))**2

      enddo



      if (flag .eq. 1) then

         call results(nfit,*310)

         write(luw,*) '***** Error linking with IGV *****'

         call mystop(1001)

         return 1

310   continue

      endif



      end

�Subroutine DRYCOIL





	subroutine drycoil (time,info,ma,mw,Tdbi,Twbi,Twi,C1,C2,C3,

     $                      C4,C5,C6,Q,Tdbo,Twbo,Two,dPa,dPw,

     $                      cpa,cpw,turb,liqtype,liqcomp,Tso,*) 

***********************************************************************

*  Subroutine DRYCOIL                                                 

*                                                                     

*  Subroutine DRYCOIL models the performance of the dry portion of 	

*  a cataloged cooling coil.  Characteristic parameters calculated    

*  by TRNFIT are required.  Variations in the heat transfer           

*  coefficients are accounted for by calculating fluid transport      

*  properties with NEWFLUIDS.						                    

*                                                                     

*  Updated 4/17/97  RJR                                               

*

***********************************************************************

	implicit none

	real psydat(9),time

	character*2 units

	integer info(15),liqtype,liqcomp,status,turb

	integer iter,maxiter,lur,luw,iform,luk,phase

	parameter (maxiter=100)

	double precision C1,C2,C3,C4,C5,C6,wi,Tsm,

     $          eff,NTU,Cmin,Cmax,UAtot,rhowm,

     $          Q,Tdbo,Twbo,cpwm,cpws,kwm,kws,

     $          muwm,muws,cpam,cpas,kam,kas,muam,muas,

     $          Pram,Pras,Prwm,Cstar,hAa,hAw,

     $          nTwo,Tdpi,Tso,

     $          rhostd,Twm,nTdbo,

     $          Tdbm,Two,Twi,ma,mw,Twbi,Tdbi,rhoam,

     $          cpa,cpw,dPa,dPw,

     $          dPwd,dPww,wmax,hmax

	common /lunits/lur,luw,iform,luk

	external newfluids,psych       



      psydat(1) = 1.0

      psydat(2) = Tdbi

      psydat(3) = Twbi

      call psych(time,info,1,1,1,psydat,1,status,*101)

      write(luw,*) '***** Error linking with PSYCH *****'

      call mystop(1001)

      return 1

 101 continue

      wi = psydat(6)



C  DRY COIL OPERATION

C  Initial guesses for leaving water temperature and dry bulb

      Two = Twi + 5.d0 

      Tdbo = Tdbi - 20.d0

      iter = 0



150 continue

      iter=iter+1

      Tdbm = (Tdbi+Tdbo)/2.d0

      Twm  = (Twi+Two)/2.d0

      Tsm  = (Tdbm+Twm)/2.d0



C  Liquid properties

      call newfluids (liqtype,liqcomp,Twm,Tsm,cpwm,cpws,kwm,kws,

     $               muwm,muws,rhowm,phase,*151)

      write(luw,*) '***** Error linking with NEWFLUIDS *****'

      call mystop(1001)

      return 1

 151 continue



C  Air properties

      call newfluids (2,0,Tdbm,Tsm,cpam,cpas,kam,kas,muam,

     $                 muas,rhoam,phase,*152)

 152 continue



C  Prandtl numbers

      Pram=muam*cpam/kam*1000

      Pras=muas*cpas/kas*1000   

      Prwm=muwm*cpwm/kwm*1000



C  UA products

C  Air

      hAa=C1*(ma/muam)**C2*Pram**0.36*

     $         (Pram/Pras)**0.25*kam



C  Water

      if (turb .eq. 1) then

         hAw=C3*(mw/muwm)**0.7*Prwm**0.33*

     $         (muwm/muws)**0.14*kwm

      else

         hAw=C3*(mw/muwm)**0.8*Prwm**0.33*

     $         (muwm/muws)**0.14*kwm

      endif



      UAtot=1.d0/(1.d0/hAa+1.d0/hAw)



C  Capacitance rates

      Cmin=min(ma*cpa,mw*cpw)

      Cmax=max(ma*cpa,mw*cpw)

      Cstar=Cmin/Cmax



C  Number of transfer units

      NTU=UAtot/Cmin



C  Counterflow effectiveness

      eff=(1.d0-exp(-NTU*(1.d0-Cstar)))/

     $      (1.d0-Cstar*exp(-NTU*(1.d0-Cstar)))



C  Heat Transfer

      Q=eff*Cmin*(Tdbi-Twi)



C  Outlet States

      nTdbo=Tdbi-Q/(ma*cpa)

      nTwo =Twi+Q/(mw*cpw)



C  Convergence

      if ((abs(Tdbo-nTdbo)/max(1.d0,Tdbo) .ge. 0.01) .or.

     $    (abs(Two -nTwo )/max(1.d0,Two ) .ge. 0.01)) then

         if (iter .lt. maxiter) then 

            Tdbo = nTdbo

            Two  = nTwo

            goto 150

         endif

      endif



      psydat(2)=Tdbo

      psydat(4)=1.d0

      call psych(time,info,1,2,1,psydat,2,status,*102)

 102 continue

      wmax=psydat(6)



      psydat(6)=min(wi,wmax)

      call psych(time,info,1,4,1,psydat,2,status,*103)

 103 continue

      Twbo = psydat(3)



      Tso=Tdbo+(Twi-Tdbo)*(hAa/(hAa+hAw))



      dPa = C4*(ma**2/rhoam)*(ma/muam)**C5

      dPw = C6*(mw/muwm)**(-0.25)*(mw**2/rhowm)*

     $            (muws/muwm)**(-0.14)



      return 1

      end

  �Subroutine WETCOIL





	subroutine wetcoil (time,info,ma,mw,Tdbi,Twbi,Twi,C1,C2,C3,

     $                      C4,C5,C6,QT,QS,QL,Tdbo,Twbo,Two,dPa,dPw,

     $                      cpa,cpw,Aface,Tsi,turb,liqtype,liqcomp,*)

***********************************************************************

*  Subroutine WETCOIL                                                 

*                                                                     

*  Subroutine WETCOIL models the performance of the wet portion of 	

*  a cataloged cooling coil.  Characteristic parameters calculated    

*  by TRNFIT are required.  Variations in the heat transfer           

*  coefficients are accounted for by calculating fluid transport      

*  properties with NEWFLUIDS.						                    

*                                                                     

*  Updated 4/17/97  RJR                                               

*

***********************************************************************

	implicit none

	real time

	real psydat(9)

	character*2 units

	integer info(15),turb,lur,luw,iform,luk

	integer liqtype,liqcomp,status

	integer iter,maxiter,phase

	parameter (maxiter=100)

	double precision C1,C2,C3,C4,C5,C6,wi,Tsm,

     $          eff,NTU,rhowm,

     $          QS,QL,QT,Tdbo,Twbo,cpwm,cpws,kwm,kws,

     $          muwm,muws,cpam,cpas,kam,kas,muam,muas,

     $          Pram,Pras,Prwm,Cstar,hAa,hAw,

     $          nTwo,Tdpi,hTdpi,hTwi,cpsat,

     $          hai,hwsi,rhostd,Vstd,Cf,UAhtot,capAirWet,capLiqWet,

     $          mmin,mmax,mstar,hwso,Ri,hao,Rrat,Ro,hssi,hsso,

     $          Tsi,hnew,airCap,cpv,Tcond,hscond,

     $          wo,Qmax,hx,nTdbo,Twm,Tdbm,

     $          Two,Twi,ma,mw,Twbi,Tdbi,rhoam,

     $          cpa,cpw,Aface,Tdwo,dPa,dPw,

     $          wmax,hmax

	parameter (cpv=1.87,rhostd=1.184)

	common /lunits/lur,luw,iform,luk

	external newfluids,psych 



      psydat(1) = 1.0

      psydat(2) = Tdbi

      psydat(3) = Twbi

      call psych(time,info,1,1,1,psydat,1,status,*101)

      write(luw,*) '***** Error linking with PSYCH *****'

      call mystop(1001)

      return 1 

 101 continue

      wi = psydat(6)



      iter=0



C  WET COIL OPERATION

C  Calculate entering dewpoint, make guess about leaving

C  dry bulb temperature and leaving liquid temperature 

      if (Tdbi .eq. Twbi) then

         Tdpi=Tdbi

      else

         psydat(2) = Tdbi

         psydat(3) = Twbi

         call psych(time,info,1,1,1,psydat,2,status,*104)

 104   continue

         Tdpi = psydat(5)

      endif

      Tdbo = Tdbi - 20.d0

      Two = Twi + 5.d0



C  Mean temperatures

 150	continue

      iter=iter+1

      Tdbm = (Tdbi+Tdbo)/2.d0

      Twm = (Twi+Two)/2.d0

      Tsm = (Tdbm+Twm)/2.d0



C  Saturation enthalpy at entering dewpoint

      psydat(2)=Tdpi

      psydat(4)=1.d0

      call psych(time,info,1,2,1,psydat,2,status,*151)

 151 continue

      hTdpi=psydat(7)



C  Saturation enthalpy at entering liquid temperature

      psydat(2)=Twi

      call psych(time,info,1,2,1,psydat,2,status,*152)

 152 continue

      hTwi=psydat(7)



C  Effective specific heat of saturated air

      cpsat=max((hTdpi-hTwi),1.d-5)/max((Tdpi-Twi),1.d-5)



C  Air enthalpy at inlet conditions

      psydat(2) = Tdbi

      psydat(3) = Twbi

      call psych(time,info,1,1,1,psydat,2,status,*153)

 153 continue

      hai=psydat(7)



C  Saturation enthalpy of air at inlet liquid temperature

      psydat(2) = Twi

      psydat(4) = 1.d0

      call psych(time,info,1,2,1,psydat,2,status,*154)

 154 continue

      hwsi=psydat(7)



C  Liquid properties

      call newfluids (liqtype,liqcomp,Twm,Tsm,cpwm,cpws,

     $               kwm,kws,muwm,muws,rhowm,phase,*160)

      write(luw,*) '***** Error linking with NEWFLUIDS *****'

      call mystop(1001)

      return 1

160 continue



C  Air properties

      call newfluids (2,0,Tdbm,Tsm,cpam,cpas,kam,kas,muam,

     $                  muas,rhoam,phase,*165)

165 continue 



C  Prandtl numbers

      Pram=muam*cpam/kam*1000

      Pras=muas*cpas/kas*1000

      Prwm=muwm*cpwm/kwm*1000



C  Wet surface convection coefficient correction factor

      Vstd=ma/(rhostd*Aface)/60.0/0.3048

      Cf=0.626*Vstd**0.101



C  Conductance-heat transfer area products

      hAa=Cf*C1*(ma/muam)**C2*Pram**0.36*

     $            (Pram/Pras)**0.25*kam

      

      if (turb .eq. 1) then

         hAw=   C3*(mw/muwm)**0.7*Prwm**0.33*

     $            (muwm/muws)**0.14*kwm

      else 

         hAw=   C3*(mw/muwm)**0.8*Prwm**0.33*

     $            (muwm/muws)**0.14*kwm

      endif 

       

C  Overall enthalpy transfer coefficient       

      UAhtot=1.d0/(cpsat/hAw+cpam/hAa)



C  Air-side and liquid-side capacity rates

      capAirWet=ma

      capLiqWet=mw*cpw/cpsat



C  Enthalpy-based capacitance rates

      mmin=min(capAirWet,capLiqWet)

      mmax=max(capAirWet,capLiqWet)



      mstar=mmin/mmax



C  Number of transfer units

      NTU=UAhtot/mmin



C  Counterflow effectiveness

      eff=(1.d0-exp(-NTU*(1.d0-mstar)))/

     $      (1.d0-mstar*exp(-NTU*(1.d0-mstar)))



C  Maximum heat transfer rate

      Qmax=mmin*abs(hai-hwsi)



C  Air enthalpy at outlet

      hao=hai-eff*Qmax/capAirWet        



C  Saturation enthalpy of air at exit liquid temperature

      hwso=hwsi+eff*Qmax/capLiqWet



C  Heat transfer resistances

      Ri=1.d0/hAw

      Ro=1.d0/hAa

      Rrat=Ri/(Ri+cpam/cpsat*Ro)



C  Saturation enthalpy of air at entering and exit 

C  surface temperature      

      hssi=hwso+Rrat*(hai-hwso)

      hsso=hwsi+Rrat*(hao-hwsi)



C  Total heat transfer rate for a totally wet coil

      QT=ma*(hai-hao)



C  New calculated leaving liquid temperature

      nTwo=Twi+QT/(mw*cpw)



C  Initial guess for surface temperature at air inlet

      Tsi=nTwo



C  Determine surface temperature corresponding to previously

C  calculated saturation enthalpy of air at entering

C  surface temperature

 205 continue

      psydat(2) = Tsi 

      psydat(4) = 1.d0

      call psych(time,info,1,2,1,psydat,2,status,*206)

 206 continue

      hnew=psydat(7)

      if (hnew .lt. hssi) then

         Tsi=Tsi + 0.1

         goto 205

      endif

   

C  Determine temperature effectiveness of air stream

      airCap=ma*(cpa+wi*cpv)

      NTU=hAa/airCap

      eff=1.d0-exp(-NTU)

    

C  Enthalpy of condensate

      hscond=hai-(hai-hao)/eff

    

C  First guess of condensate temperature

      Tcond = Twi



C  Determine condensate temperature corresponding to

C  condensate enthalpy

 207 continue

      psydat(2) = Tcond

      psydat(4) = 1.d0

      call psych(time,info,1,2,1,psydat,2,status,*208)

 208 continue

      hnew=psydat(7)

      if (hnew .lt. hscond) then

         Tcond=Tcond+0.1

         goto 207

      endif



C  New calculated leaving dry bulb temperature for a wet coil

      nTdbo=Tdbi-(Tdbi-Tcond)*eff



C  Check convergence of leaving dry bulb temperature and

C  leaving liquid temperature

      if ((abs(Tdbo-nTdbo)/max(1.d0,Tdbo) .ge. 0.01) .or.

     $    (abs(Two -nTwo )/max(1.d0,Two) .ge. 0.01)) then

         if (iter .lt. maxiter) then

            Tdbo=nTdbo

            Two=nTwo

            goto 150

         endif

      endif

      

C  Calculate leaving wet bulb temperature and leaving   

C  humidity ratio

      psydat(2)=Tdbo

      psydat(4)=1.e0

      call psych(time,info,1,2,1,psydat,2,status,*209)

 209 continue

      hmax=psydat(7)

      

      psydat(7)=min(hao,hmax)

      call psych(time,info,1,5,1,psydat,2,status,*210)

 210 continue

      Twbo=psydat(3)

     

      if (Twbo .gt. Tdbo) then

         Twbo=Tdbo

         psydat(2)=Tdbo

         psydat(3)=Tdbo

         call psych(time,info,1,1,1,psydat,2,status,*211)          

 211  continue

      endif



      wo=psydat(6)

     

C  Calculate enthalpy at fictitious state x

      psydat(2)=Tdbi

      psydat(6)=min(wi,wo)

      call psych(time,info,1,4,1,psydat,2,status,*212)

 212 continue

      hx=psydat(7)



C  Calculate latent and sensible heat transfers

      QL=ma*(hai-hx)

      QS=ma*(hx-hao)

     

      dPa = C4*(ma**2/rhoam)*(ma/muam)**C5

      dPw = C6*(mw/muwm)**(-0.25)*(mw**2/rhowm)*

     $            (muws/muwm)**(-0.14)



      return 1 

      end

